Something for Nothing! Converting Plots from SAS/GRAPH® to ODS Graphics

Philip R Holland, Holland Numerics Ltd, Royston, Hertfordshire, United Kingdom

ABSTRACT
All the documentation about the creation of graphs with SAS® software states that ODS Graphics is not intended to replace SAS/GRAPH®. However, ODS Graphics is included in the Base SAS® licence from SAS® 9.3, but SAS/GRAPH still requires an additional component licence, so there is definitely a financial incentive to convert to ODS Graphics. This paper gives examples that can be used to replace commonly created SAS/GRAPH plots, and highlights the small number of plots that are still very difficult, or impossible, to create in ODS Graphics.

INTRODUCTION
The paper will compare the output from the most commonly used SAS code for producing plots and charts using SAS/GRAPH and ODS Graphics. Each ODS Graphics program will be rated as follows:

- Easy: Replacing one SAS/GRAPH procedure statement with an ODS Graphics procedure. The conversion may also require some pre-processing of the input data.
- Difficult: Replacing one SAS/GRAPH procedure with code containing PROC TEMPLATE and PROC SGRRENDER, or the conversion requires extensive pre-processing of the input data.
- Impossible: There is currently no corresponding ODS Graphics procedure in that version of SAS software to replicate the SAS/GRAPH graph.

SCATTER PLOTS
The first and simplest of the commonly used plots is the scatter plot. Basically a lot of data points scattered over the graph area.

SAS/GRAPH FROM SAS 9.2

```sas
PROC GPLOT DATA = sashelp.class;
SYMBOL V = CIRCLE I = NONE;
PLOT weight * height = sex;
RUN;
```

ODS GRAPHICS FROM SAS 9.2 (EASY)

It should be noted that, because the input data is unsorted, and the first record contains a record where sex='M', the legend starts with sex='M'.

```sas
PROC SGPLOT DATA = sashelp.class;
  VScatter x=height y=weight; 
  RUN;
```

The first record contains a record where sex='M', the legend starts with sex='M'.

```sas
PROC SGPLOT DATA = sashelp.class;
  VScatter x=height y=weight; 
  RUN;
```
LINE PLOTS
The second and probably the most frequently used of the common plots is the line plot. SAS/GRAPH provides a vast range of options as to how the individual points are used to create the line, but the sample code is just joining each point to the next, which is why the points have been sorted by the values on the x-axis.

SAS/GRAPH FROM SAS 9.2

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex height;
RUN;

PROC GPLOT DATA = class;
   SYMBOL V = CIRCLE I = JOIN;
   PLOT weight * height = sex;
RUN;

ODS GRAPHICS FROM SAS 9.2 (EASY)
It should be noted that, now the input data has been sorted, and the first record contains a record where sex='F', so the legend order will match that in the SAS/GRAPH plot. There are also fewer ticks on the 2 axes.
**REGRESSION PLOTS**

Another way to use the points to generate a line is to calculate a regression line and confidence limits.

**SAS/GRAPH FROM SAS 9.2**

In SAS/GRAPH the regression and confidence limit lines are plotted well beyond the limits of the data points, and extend to the edge of the plotting area.

```
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex height;
RUN;

PROC SGPLOT DATA = class;
   SERIES Y = weight X = height / GROUP = sex MARKERS
   MARKERATTRS = (SYMBOL=CIRCLE);
RUN;
```

**ODS GRAPHICS FROM SAS 9.2 (EASY)**

Again there are fewer ticks on the axes, and the minimum and maximum values on the axes are different to accommodate all of the confidence limit lines, whereas the SAS/GRAPH axis ranges are based on the actual data points. Unlike in the SAS/GRAPH plot, the confidence limit lines in the ODS Graphics plot do not extend beyond the range of the data values on the x-axis, so no data extrapolation is being carried out.

```
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex height;
RUN;

PROC GPLOT DATA = class;
   SYMBOL V = CIRCLE I = R0CLI95;
   PLOT weight * height = sex;
RUN;
```
ERROR BAR PLOTS

This type of plot is often created using a scatter plot and a fair amount of annotation, but it is still possible in SAS/GRAPH using a little pre-processing of the input data to change the single data point into three points: original, upper error limit and lower error limit.

SAS/GRAPH FROM SAS 9.2

```sas
PROC SORT DATA = sashelp.class
OUT = class;
BY sex height;
RUN;

PROC SUMMARY DATA = class NWAY;
CLASS sex;
VAR weight;
OUTPUT OUT = class_se
STDERR = weight_se;
RUN;

DATA class_classic
(KEEP = sex height value);
MERGE class class_se;
BY sex;
value = weight;
OUTPUT;
value = weight + weight_se;
OUTPUT;
value = weight - weight_se;
OUTPUT;
RUN;

PROC GPLOT DATA = class_classic;
SYMBOL V = CIRCLE I = HILOTJ;
PLOT value * height = sex;
LABEL value = "Weight";
RUN;
```

ODS GRAPHICS FROM SAS 9.2 (EASY)

ODS Graphics also requires a little pre-processing of the input data to add two extra values to each data point for the upper and lower error limits. This plot demonstrates an important feature of ODS Graphics, where graphical elements can be drawn on top of the previously drawn elements. In this case the error bars are drawn using the SCATTER statement, then the line drawn on top using the SERIES statement, so the error bars appear behind the line.
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex height;
RUN;
PROC SUMMARY DATA = class NWAY;
   CLASS sex;
   VAR weight;
   OUTPUT OUT = class_se
       STDERR = weight_se;
RUN;
DATA class_ods
   (KEEP = sex height value
       value_upper value_lower);
MERGE class class_se;
   BY sex;
   value = weight;
   value_upper = weight + weight_se;
   value_lower = weight - weight_se;
OUTPUT;
RUN;
PROC SGPLOT DATA = class_ods;
   SCATTER Y = value X = height /
       GROUP = sex
       YERRORUPPER = value_upper
       YERRORLOWER = value_lower;
   SERIES Y = value X = height /
       GROUP = sex;
   LABEL value = "Weight";
RUN;

**BOX PLOTS**
The box plot can be used to display the simple statistics, including the quartiles and outliers.

**SAS/GRAFGRAPH FROM SAS 9.2**
The default SAS/GRAFGRAPH plot certainly requires a number of features to be customized, such as the box width, to make it useful for the viewer.
ODS GRAPHICS FROM SAS 9.2 (EASY)

Using the default settings for ODS Graphics produces a box plot that is recognizably similar to that produced using the default settings in SAS/GRAPH, but where the viewer can see much more clearly the information in the graph.

PROC SORT DATA = sashelp.class
   OUT = class;
   BY age;
RUN;

PROC SGPLOT DATA = class;
   VBOX height / CATEGORY = age;
RUN;

VERTICAL BAR CHARTS

This is the first group of bar charts created by PROC GCHART in SAS/GRAPH. Each group will include a simple bar chart, a stacked bar chart (known as “sub-grouped” in SAS/GRAPH) and a clustered bar chart (known as “grouped” in SAS/GRAPH). All of the graphs in this paper are drawn in square graph areas.

SAS/GRAPH FROM SAS 9.2

The simple and stacked bar charts drawn by SAS/GRAPH appear unusually narrow, probably because the default bar widths are fixed, rather than adapting to the graph area.
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   VBAR age / DISCRETE;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   VBAR age / SUBGROUP = sex
   DISCRETE;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   VBAR sex / GROUP = age
   PATTERNID = MIDPOINT;
RUN;
Vertical bar charts are available in the basic and stacked forms from SAS 9.2, where "stacked" is the new name for "sub-grouped". The clustered form can only be approximated by using the panelled form in PROC SGPANEL. Note that the graph bars make full use of the available graph area!

```sas
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC SGPLOT DATA = class;
   VBAR age;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC SGPLOT DATA = class;
   VBAR age / GROUP = sex;
RUN;
```
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC SGPANEL DATA = class;
   PANELBY age / COLUMNS = 6;
   VBAR sex / GROUP = sex;
RUN;

**ODS GRAPHICS FROM SAS 9.3 (EASY)**
From SAS 9.3 the clustered vertical bar chart is supported using the GROUPDISPLAY=CLUSTER option, so the panelled form is no longer necessary. However, the identification of the bars is solely with the legend, as it is for the stacked bar chart.

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;
PROC SGPLOT DATA = class;
   VBAR age / GROUP = sex
   GROUPDISPLAY = CLUSTER;
RUN;

**HORIZONTAL BAR CHARTS**
The default horizontal bar chart in SAS/GRAPH includes statistics at the right-hand side of the chart. This feature creates the biggest problem when attempting to replicate this chart in ODS Graphics.

**SAS/GRAPH FROM SAS 9.2**
Note that the default horizontal bar charts, unlike the vertical bar charts, make full use of the available graph area.
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   HBAR age / DISCRETE;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   HBAR age / SUBGROUP = sex
   DISCRETE;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   HBAR age / GROUP = age
   PATTERNID = MIDPOINT;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC GCHART DATA = class;
   HBAR sex / SUBGROUP = age
   DISCRETE;
RUN;
ODS GRAPHICS FROM SAS 9.2 (EASY)

If you just want to create a horizontal bar chart in the form of a "rotated" vertical bar chart, then this is certainly possible in SAS 9.2 for the basic and stacked forms, and are shown below. However, adding the statistics on the right-hand side of the bar chart will not be created here, as they require extensive PROC TEMPLATE code to plot text versions of the statistics in separate layouts.

```
PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC SGPLOT DATA = class;
   HBAR age;
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC SGPLOT DATA = class;
   HBAR age / GROUP = sex;
RUN;
```
ODS GRAPHICS FROM SAS 9.3 (EASY)

From SAS 9.3 the clustered horizontal bar chart is supported using the GROUPDISPLAY=CLUSTER option, so the panelled form is no longer necessary. However, the identification of the bars is solely with the legend, as it is for the stacked bar chart.

```
PROC SORT DATA = sashelp.class
    OUT = class;
    BY sex age;
RUN;

PROC SGPLOT DATA = class;
    HBAR age / GROUP = sex
        GROUPDISPLAY = CLUSTER;
RUN;
```

ODS GRAPHICS FROM SAS 9.4 (DIFFICULT AND IMPOSSIBLE)

From SAS 9.4 the YAXISTABLE statement in PROC SGPLOT allows the alignment of text and bars, but only for the primary midpoint axis in SAS 9.4 and 9.4M1, and not for the clustering axis, so only single rows of text can be included. The capability to align text with the clustering axis is expected to be included in SAS 9.4M2. As you can see below the data needs to be carefully prepared before rendering, which makes it much less convenient to use than the HBAR statement in PROC GCHART.

```
PROC SORT DATA = sashelp.class
    OUT = class;
    BY sex age;
RUN;

PROC SGPANEL DATA = class;
    PANELBY age / ROWS = 6;
    HBAR sex / GROUP = sex;
RUN;
```
PROC SORT DATA = sashelp.class
   OUT = class;
   BY age;
RUN;

PROC FREQ DATA = class;
   TABLES age /
       OUT = class_summ NOPRINT;
RUN;

DATA class_summ;
   SET class_summ;
   BY age;
   RETAIN cum_freq cum_pct .;
   freq = count;
   cum_freq + count;
   cum_pct + percent;
   FORMAT freq cum_freq 3.
       percent cum_pct 6.2;
   LABEL freq = 'FRQ'
       cum_freq = 'CFRQ'
       percent = 'PCT'
       cum_pct = 'CPCT';
RUN;

PROC SORT DATA = sashelp.class
   OUT = class;
   BY sex age;
RUN;

PROC FREQ DATA = class;
   TABLES sex * age /
       OUT = class_stack_summ NOPRINT;
RUN;

DATA class_stack_summ;
   SET class_stack_summ;
   BY sex age;
   RETAIN cum_freq cum_pct .;
   freq = count;
   cum_freq + count;
   cum_pct + percent;
   FORMAT freq cum_freq 3.
       percent cum_pct 6.2;
   LABEL freq = 'FRQ'
       cum_freq = 'CFRQ'
       percent = 'PCT'
       cum_pct = 'CPCT';
RUN;

DATA class_stack_summ;
   SET class_stack_summ;
   BY sex age;
   IF NOT LAST.age THEN DO;
      freq = .;
      cum_freq = .;
      percent = .;
      cum_pct = .;
   END;
RUN;

PROC SG PLOT DATA = class_summ;
   HBAR age / STAT = FREQ;
   YAXISTABLE freq cum_freq
       percent cum_pct /
       LOCATION = INSIDE
       POSITION = RIGHT
       LABELPOS = TOP;
RUN;

PROC SG PLOT DATA = class_stack_summ;
   HBAR age / GROUP = sex;
   YAXISTABLE freq cum_freq
       percent cum_pct /
       LOCATION = INSIDE
       POSITION = RIGHT
       LABELPOS = TOP;
RUN;
**2D PIE CHARTS**

2D pie charts are simple to create using SAS/GRAPH, and there are many options to manipulate the way the segments and labels are presented. The sample code used has been deliberately kept very simple.

**SAS/GRAPH FROM SAS 9.2**

```
PROC SORT DATA = sashelp.class
  OUT = class;
  BY age;
RUN;

PROC GCHART DATA = class;
  PIE age / VALUE = ARROW
    LEGEND DISCRETE;
RUN;
```

**ODS GRAPHICS IN SAS 9.2 (IMPOSSIBLE)**

2D pie charts are not available in the SG procedures, nor in PROC TEMPLATE, in SAS 9.2.

**ODS GRAPHICS FROM SAS 9.3 (DIFFICULT)**

2D pie charts are available in PROC TEMPLATE, but not in the SG procedures, from SAS 9.3. The ODS Graphics sample has been deliberately plotted with the segments in a clockwise direction from 9:00, instead of anti-clockwise from 3:00 as in the SAS/GRAPH code, to emphasize the differences.

```
PROC SORT DATA = sashelp.class
  OUT = class;
  BY age;
RUN;

PROC TEMPLATE;
  DEFINE STATGRAPH pie;
  BEGINGRAPH;
    LAYOUT REGION;
      PIECHART CATEGORY = age /
        DATALABELLOCATION = OUTSIDE
        CATEGORYDIRECTION = CLOCKWISE
        START = 180
        NAME = 'pie';
      DISCRETELEGEND 'pie' /
        TITLE = 'Age';
    ENDLAYOUT;
  ENDPGRAPH;
END;
RUN;

PROC SGRENDER DATA = class
  TEMPLATE = pie;
RUN;
```
3D PIE CHARTS
This chart is one that I have never really liked. My opinion of this graph is that, rather than being a way to inform the viewer, they are frequently used to mislead the viewer instead. As a consequence I have not included any SAS/GRAPH sample code here.

ODS GRAPHICS IN SAS 9.2, 9.3 AND 9.4 (IMPOSSIBLE)
It is with great delight that I have discovered that 3D pie charts will not be supported in any release of ODS Graphics in the near future. This does not prevent SAS users who wish to create these charts from doing so, but they will be required to license SAS/GRAPH first.

CONCLUSIONS
If you are reading this paper with a view to converting all of your SAS graphics program from SAS/GRAPH to ODS Graphics, then you must consider the following questions:

• Do you only create plots, and no charts at all? This probably means that your programs could be converted from SAS/GRAPH to ODS Graphics in SAS 9.2 or 9.3, provided any annotation currently used can be plotted as another plot to be overlaid over the base plot.

• If you create bar charts, do you draw horizontal bar charts that require large amounts of text added alongside the chart? If the bar charts, whether vertical or horizontal, only require a single value plotted in or at the end of the bar, then it should still be possible to convert your programs from SAS/GRAPH to ODS Graphics in SAS 9.2 or 9.3, although some PROC TEMPLATE code may be necessary for the more complex charts.

• If you require horizontal clustered bar charts with large amounts of text added, 2D or 3D pie charts, then it may not be possible to convert your SAS/GRAPH programs to ODS Graphics, even in SAS 9.4, or, at best, only possible with a great deal of effort. It is recommended, therefore, that you keep your SAS/GRAPH licence for now, as PROC GCHART is not going away in the foreseeable future!

ACKNOWLEDGEMENTS
I would like to thank Sanjay Matange for his invaluable assistance in resolving a number of programming issues in ODS Graphics.

RECOMMENDED READING


CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
Name: Philip R Holland
Organization: Holland Numerics Limited
Address: 94 Green Drift, Royston, Hertfordshire, SG8 5BT, United Kingdom
Work Phone: +44-7714-279085
Email: phil@hollandnumerics.com
Web: www.hollandnumerics.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.